



36

RANDOM WALKS: MEASURING DISPERSAL SUCCESS

Objectives

- Set up a spreadsheet model that simulates a random landscape.
- Simulate dispersal of an organism across the random landscape.
- Write a macro to calculate dispersal success over many trials.
- Determine the relationship between dispersal success, mortality threat in crossing hostile habitat, and proportion of the landscape in forested habitat.
- Evaluate how maximum dispersal distances affect dispersal success or failure.

Prerequisite Exercise: Landscape Ecology: Quantifying Landscape Patterns

INTRODUCTION

Dispersal is a critical part of the life cycle for many organisms. In a broad sense, any movement of an organism from one location to another constitutes a dispersal event. In a more restricted sense, dispersal is the movement of an organism from its birth location to another location at which it reproduces. For example, young birds and mammals disperse from their natal birthplaces to begin a reproductive life that is independent from their parents. The migrations of monarch butterflies and migratory birds also constitute dispersal movements, but these movements may or may not affect the structure of a population (see Exercise 22, “Gene Flow and Population Structure”).

Plants disperse too, although they cannot physically pick up their roots and move to a new location. In plants, dispersal of gametes occurs through the action of wind or pollinators that move pollen from location to location. Plant offspring (seeds) can be dispersed by a variety of means, including wind or rain action or through an agent that moves the seed from one location to another, such as a seed-eater that consumes seeds in one location and deposits them in new locations.

Why do organisms disperse? From an evolutionary standpoint, organisms may disperse away from their natal locations to “avoid” inbreeding, or to decrease density-dependent mortality. In short, dispersal may have evolved because those individuals that dispersed had higher fitnesses than those individuals that did not. Ecologically, organisms may disperse from location to location to find food, water, mates, or shelter. If you have studied island biogeography, gene flow, sources and

sinks, metapopulations, or range expansion in this book, you know that dispersal of individuals shapes species' ranges and plays a central role in the ecology and evolution of populations.

But dispersal can be costly. Organisms often encounter novel ecological conditions when they disperse, which can decrease their probability of survival. For example, a newly paved road through a forest can severely affect the probability that a woodland frog can successfully disperse to and from its breeding pond. If a large number of frogs fail to navigate successfully to the breeding pond, the population may decline or even become extinct. Thus, the structure and composition of a landscape can impact the ability of populations to persist.

In this exercise, you'll develop a model to investigate how landscape composition affects the dispersal success of a hypothetical organism, such as a woodland frog. We will assume that our hypothetical frog is attempting to disperse across the landscape to a target location, such as a breeding pond. The habitat in the landscape will be randomly assigned to cells. You'll set your landscape to a certain composition (e.g., 90% woodland habitat and 10% nonwoodland habitat), but the arrangement of the forest on the landscape will be randomly determined. You constructed such a landscape in Exercise 30, our introduction to landscape ecology. We will assume that our hypothetical woodland frog can cross nonwoodland habitat, but faces some mortality risks in utilizing this habitat type (for example, dehydration or increased predation). This mortality risk will be an important part of your model.

We will give our frog a set of rules for determining which cell it will move to; these rules are diagrammed and described in Figure 1. Because our hypothetical woodland frog has an **algorithm**—an unambiguous set of rules for making each move—we can calculate the probability that it will make it safely across a random landscape given the following:

- the proportion of the landscape that is safe woodland habitat (H)
- the risk associated with crossing nonwoodland habitat (R)
- the number of moves the frog needs to make to get across (n)

To calculate this probability, we need first to work out the probability that the frog will be forced into nonwoodland habitat on any given move. Recall that when the frog

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1	Dispersal "rules"													
2			Possible movements					Landscape						
3														
4														
5														
6														
7					Start					Start				
8														

Figure 1 Let's suppose a woodland frog situated on the habitat in cell E7 must traverse the landscape of cells D3–F6. Its destination is somewhere on the landscape in row 3. We will assume that our organism can choose to move into any of the shaded cells (cells D6–F6)—that is, it will always move one row up, and it can choose from the three cells closest to its current position in that row. Suppose now that our frog is situated at cell J7 and needs to cross a landscape (cells I3–K6) that consists of shaded cells (woodland habitat) and nonshaded cells (nonwoodland habitat). We can see that our frog can move from cell J7 to J6 to I5 to J4 to K3 (the breeding pond) without crossing potentially hostile nonwoodland habitat. This would be considered a "successful" dispersal movement. If, however, all three of the cells ahead of the frog at any step were nonwoodland habitat, the frog would have to take the risk of crossing nonwoodland habitat to proceed toward the pond.

makes a move, it looks at three cells one row ahead and will only move into non-woodland habitat if none of those cells is woodland habitat. Each cell in the landscape has probability $1 - H$ of being nonwoodland habitat, so the probability that *all* three cells ahead of the frog will be nonwoodland habitat is $(1 - H)^3$, and this is also the probability that the frog will be forced to move into nonwoodland habitat on any given move. In general, if an organism can consider x different cells, the probability that it will be forced into nonwoodland habitat is $(1 - H)^x$.

We now need to calculate the probability that the frog will survive one move, $p(1)$. It is actually easier to approach this problem by calculating the probability that the frog will *not* survive a given move: this is the probability that it will be forced to enter nonwoodland habitat, which we just saw was $(1 - H)^3$, times the risk it takes in leaving woodland habitat (R). Since each move has only two possible outcomes (the frog survives or it doesn't), and since these two outcomes are mutually exclusive, the total probability for both outcomes will be 1. The probability that the frog survives the move will therefore be 1 minus the probability that it does not survive:

$$p(1) = 1 - R(1 - H)^3$$

The probability that the frog will survive n moves across a random landscape, $p(n)$, is therefore

$$p(n) = [1 - R(1 - H)^3]^n \tag{Equation 1}$$

Figure 2 shows how the probability of successfully crossing a random landscape varies when we hold the risk of crossing nonwoodland habitat constant at 0.3 and let the proportion of woodland habitat vary from 0 to 1. The graph shows two cases: when the number of steps required to cross (n) is 10, and when the number of steps required to cross is 100. Both cases show that the probability of safely crossing is a non-linear function of the amount of woodland habitat in the landscape.

It is also possible to predict, on average, how many moves a frog will be able to make before it dies, though the mathematics behind this is a little more complicated. Suppose we were to send our virtual frog across a very large landscape many times (or send many frogs across a very large landscape once each), track the number of moves each frog




Figure 2 The probability of crossing a random landscape safely when the risk of crossing nonwoodland habitat, R , is 0.3, and the proportion of woodland habitat varies from 0 to 1.




Figure 3 The expected number of moves an organism can make across a random landscape when the risk of crossing nonwoodland habitat, R , is 0.3, and the proportion of woodland habitat varies from 0 to 0.9.

made before dying, and calculate the mean of all those values. We would expect the result (which we call E for *expected value*) to be

$$E = \frac{1}{R(1-H)^3} \quad \text{Equation 2}$$

Figure 3 shows how this value changes when we hold the risk associated with crossing nonwoodland habitat constant at 0.3 and let the proportion of woodland habitat vary from 0 to 0.9 (this function is undefined when $H = 1$; when the entire landscape is woodland, the frog can theoretically make an infinite number of moves without dying). Note that the expected number of moves is graphed on a logarithmic axis, so the function drops off much more rapidly than it might at first appear; for example, the expected number of moves until death when $H = 0.9$ is over 3,000, but when $H = 0.5$ it is only about 27.

PROCEDURES

In this exercise, you'll develop a model to investigate how landscape composition affects the dispersal success of a hypothetical organism. The use of random (or neutral) landscapes will allow us to explore how landscape composition affects dispersal, and how risks in utilizing nonhabitat affect dispersal success. As always, save your work frequently to disk.

INSTRUCTIONS

A. Set up the landscape.

1. Open a new spreadsheet and set up headings as shown in Figure 4.
2. Enter parameters shown in cells C4–C5.

ANNOTATION

We will start with an 11×11 landscape that consists of 121 cells of either “habitat” or “nonhabitat.” The habitat of interest can be anything you'd like it to be: forest, grassland, etc.; for now, we will consider it to be forest.

Enter 0.5 in cell C4.

Enter the formula `=1-C4` in cell C5.

Our landscape will initially consist of 50% habitat and 50% nonhabitat. In cell C5, we used the formula `=1-C4` to ensure that the total proportion of the two habitats sums to 1 in case we want to change the values..

3. Enter the risk in crossing nonhabitat in cell G4.

4. Enter the starting location in cell G5.

We'll start with a mortality risk of 0.2, which means that our organism will have a 20% chance of dying when it must cross nonforest to disperse. Later in the exercise, you can change this mortality risk to a larger or smaller number.

	A	B	C	D	E	F	G	H	I	J	K	L
1	Random Walks											
2												
3	Parameters											
4	Prop habitat		0.5	Nonhabitat mortality factor		0.2						
5	Prop nonhabitat		0.5	Starting location		\$F\$20						
6												
7												
8	Random landscape											
9												
10												
11												
12												
13												
14												
15												
16												
17												
18												
19												
20						Start						
21												

Figure 4

The starting location gives the initial position of our organism, which we will refer to as a frog, on the landscape. Our frog will start on cell F20, and its intended destination is the top of the landscape (row 8), where a breeding pond occurs.

5. In cells A9–K19, enter a formula to generate a random number between 0 and 1.

Enter the formula `=RAND()` in cell A9. Copy this formula down and across to cell K19. The **RAND** function returns a random number between 0 and 1. By entering a random number into each pixel in our landscape, we can generate a random landscape. Note that when you press F9, the calculate key, the spreadsheet generates a new set of random numbers. For 11×11 cells, this is a very small landscape, but it will get you started; later you can extend the landscape to as many cells as you wish.

6. Conditionally format cells A9–K19 in the landscape so that habitat cells are shaded and nonhabitat cells remain unshaded.

Select cells A9–K19, then go to Format | Conditional Format. A dialog box (Figure 5) will appear. Enter values to establish condition 1, which will shade your “habitat” pixels a different color from nonhabitat pixels. Enter the values shown (the cell value is between 0 and the proportion of habitat listed in cell **\$C\$4**). Now click the Format button. Go to Patterns and select a color. Press OK and your landscape will be formatted. Note that when you press F9, the random numbers in the landscape will change, and the habitat cells will automatically be reshaded.



Figure 5

7. Save your work.

B. Set up the dispersal walk.

1. Set up new headings as shown in Figure 6.

2. Set up a linear series from 1 to 12 in cells A24–A35.

3. In cell B24, enter -1. Copy this number down to cell B35.

4. In cell C24, enter a formula to determine the frog's left or right movement.

We'll let our frog make 12 dispersal steps, and then measure its dispersal success. After these 12 steps, our frog will either have reached its destination (row 8) or will have died. Row 24 will give information associated with the first dispersal step, row 25 will give information on the second dispersal step, and so on.

	A	B	C	D	E	F	G	H	I	J
22	DIRECTIONS			POSITION				HABITAT CROSSED		
23	Step	Row	Column	Value	Row	Column	Address	Habitat	Safe cross?	Steps to failure

Figure 6

Enter 1 in cell A24.

In cell A25, enter =1+A24. Copy this formula down to cell A35.

Next, we need to tell the spreadsheet how many rows and how many columns our frog will move from its current position. We'll start with the number of rows. In each dispersal step, we'll let our frog move one cell towards the top of the spreadsheet. For rows, a negative number gives the number of rows *above* the given cell and a positive number gives the number of rows *below* a given cell. Thus, the value -1 in cell B24 tells the spreadsheet that our frog will move up one row from its current location with each dispersal step. If our frog could leap ahead two rows, this number would be changed to -2. For now, we will assume that our frog moves up a single row with each dispersal step.

Enter the formula =IF(OFFSET(INDIRECT(\$G\$5),-1,0)<\$C\$4,0, IF(OFFSET(INDIRECT(\$G\$5),-1,1)<\$C\$4,1, IF(OFFSET(INDIRECT(\$G\$5),-1,-1)<\$C\$4,-1,0))) in cell C24.

This formula determines how many columns our frog will move in the first step. We will assume that our frog can move either 0 columns (straight ahead), or diagonally ahead (left or right) from its current position. Columns to the *left* of a given cell are indicated by negative numbers, and columns to the *right* of a given cell are indicated by positive numbers. Thus, we want the formula in cell C24 to return the values -1, 0, or 1 as column movements.

The formula is a very long nested IF formula, with two additional functions: **OFFSET** and **INDIRECT**. The logic behind the formula is as follows: If the cell directly ahead is forest, our frog moves straight ahead; otherwise, if the cell diagonally ahead to the right is forest, the frog moves to the right; otherwise, if the cell diagonally ahead to the left is forest, the frog moves to the left. If none of the three cells ahead is forest, it moves one cell straight ahead (to nonforest habitat) and faces a mortality risk for entering nonforest habitat. (We have the frog move straight ahead as much as possible in order to reduce the probability that it will wander outside the bounds of the landscape.)

The **OFFSET** function returns a reference for a cell that is a specified number of rows and columns from a cell of interest. It has the syntax **OFFSET(reference,rows,cols)**. For example, =**OFFSET(F20,-1,-1)** returns a reference to the cell one row above and one column to the left of cell F20—that is, cell E19. To indicate the cell above and diagonally to the right of F20, our formula would be =**OFFSET(F20,-1,1)**. However, our frog won't necessarily start in cell F20; its starting location is given in cell G5. We can use the **INDIRECT** function to correct this problem: The **INDIRECT** function returns the *contents* of a cell, so if cell G5 contains the text "F20," the formula =**OFFSET(INDIRECT(\$G\$5),-1,1)** will return the same result as the formula =**OFFSET(F20,-1,1)**.

The first part of the formula in cell C24, =**IF(OFFSET(INDIRECT(\$G\$5),-1,0)<\$C\$4,0,** tells the spreadsheet to obtain the contents of cell G5 (the starting position of our frog)

5. In cell D24, enter a formula to give the value of the cell into which the frog moved.

6. In cell E24, enter a formula to determine the row number of the frog's new position.

7. In cell F24, enter a formula to determine the column number of the frog's new position.

8. In cell G24, enter a formula to determine the cell address of the frog's new position.

9. In cell H24, enter the formula to return the word "forest" or "nonforest" to specify the habitat of the frog's new address.

10. Enter an **IF** formula into cell I24 that designates whether our frog safely dispersed or not.

and offset that cell by -1 row and 0 columns (thus, the spreadsheet evaluates cell F19). If that cell's value is less than the value in cell **\$C\$4**, we know it is forest habitat, and the formula returns the value 0 to indicate that the frog should move 0 columns (i.e., it should move straight ahead into cell F19). If cell F19 is not habitat, the spreadsheet goes through the second **IF** formula, **IF(OFFSET(INDIRECT(\$G\$5),-1,1)<\$C\$4,1**. This formula offsets the value listed in cell G5 by -1 rows and 1 column. Thus, it examines cell G19. If cell G19 is forest habitat (i.e., if its value is less than the value in cell **\$C\$4**), the frog moves into that cell (the number 1 tells the spreadsheet to offset the frog's movement by 1 column to the right). If cell G19 is not habitat, the spreadsheet goes through the third **IF** formula, **IF(OFFSET(INDIRECT(\$G\$5),-1,-1)<\$C\$4,-1,0)**, which tells the spreadsheet to look at cell E19. If this cell is forest habitat, the frog moves -1 columns during this dispersal step (one column to the left). If none of the three cells ahead is habitat, the formula returns a 0 indicating that the frog should move straight ahead into nonhabitat.

Enter the formula **=OFFSET(INDIRECT(G5),B24,C24)** in cell D24.

Now that we know the number of columns and rows our frog will move, we need to record whether this cell is forest or nonforest habitat. The formula in cell D24 tells the spreadsheet to offset the value listed in cell G5 (which gives the starting position of the frog) the number of rows given in cell B24 and the number of columns given in cell C24. The spreadsheet returns the *value* associated with the new position, which in turn reveals its habitat type.

Enter the formula **=ROW(OFFSET(INDIRECT(G5),B24,C24))** in cell E24.

Although we know where our frog is going, we need to determine the cell address of the new location so that the next dispersal step is based on the current position (rather than the starting position). This can be accomplished with the **ROW**, **COLUMN**, and **ADDRESS** formulae in the next few steps. The **ROW** formula returns the row number for a reference, while **COLUMN** returns the column number for a reference. Thus, the E24 formula returns the row number of our frog's new position.

Enter the formula **=COLUMN(OFFSET(INDIRECT(G5),B24,C24))** in cell F24.

This formula returns the column number of the frog's new position after one dispersal step.

Enter the formula **=ADDRESS(E24,F24)** in cell G24.

The **ADDRESS** formula creates a cell reference as text (similar to the entry in cell G5), given row and column numbers. Thus, the formulae in cells E24, F24, and G24 together return the new position of our frog after one dispersal step. Make sure you understand how these formulae function before you proceed.

Enter the formula **=IF(D24<\$C\$4,"forest","nonforest")** in cell H24.

Now we need to determine if our frog moved into forest or nonforest habitat. The formula in cell H24 uses an **IF** formula to return the habitat type associated with the cell or pixel that our frog dispersed to. Recall that the value in cell D24 (calculated in step 5) is the value of the cell our frog moved to in its first move. If this value is less than the value in cell **\$C\$4**, the frog moved into forest habitat, so cell H24 returns the word "forest." Otherwise, the formula returns the word "nonforest."

Enter the formula **=IF(H24="forest",1,IF(RAND()>\$G\$4,1,0))** in cell I24.

If our frog moved to forest habitat (i.e., if cell H24 is "forest"), the formula returns a 1 to indicate that the frog is still alive. If the frog moved into nonforest habitat, the formula goes through the second **IF** function: it generates a random number between 0 and 1, and if that number is *greater* than the mortality risk given in cell **\$G\$4**, the formula returns a 1 to indicate that the frog is still alive. If the random number is *less* than the mortality risk, however, the frog died and the formula returns a 0.

11. Enter a **MATCH** formula in cell J24 to reveal the number of steps until a dispersal failure (death) occurred.

12. Enter formulae in cells C25–F25 to determine the column number and position of our frog for the next dispersal step, based on the frog’s new position listed in cell G24.

13. In cell I25, enter a formula that returns a 0 if the frog died in any previous dispersal step.

14. Select cells C25–F25, and copy them down to row 35. Select cells G24–H24 and copy them down to row 35. Select cell I25 and copy it down to I35.

15. Save your work.

Enter the formula **=MATCH(0,I24:I35,0)** in cell J24.

The **MATCH** function returns the position of a value in a series of values. In our case, we are interested in determining when the first 0 occurred in column I to reveal the step number in which death occurred. (Remember that crossing into nonforest habitat does not necessarily mean death occurred.) The formula tells the spreadsheet to locate the position of the number 0 in the range of cells **I24:I35**. The 0 at the end of the formula indicates that we want the position of the *first occurrence* of the number 0. If the frog successfully dispersed for 12 consecutive steps, the spreadsheet will return the value **#N/A**, which indicates that a value is not available to be computed.

We entered the following formulae:

- C25 =**IF(OFFSET(INDIRECT(G24),-1,0)<\$C\$4,0,IF(OFFSET(INDIRECT(G24),-1,1)<\$C\$4,1,IF(OFFSET(INDIRECT(G24),-1,-1)<\$C\$4,-1,0)))**
- D25 =**OFFSET(INDIRECT(G24),B25,C25)**
- E25 =**ROW(OFFSET(INDIRECT(G24),B25,C25))**
- F25 =**COLUMN(OFFSET(INDIRECT(G24),B25,C25))**

These formulae are similar to the ones entered previously, except that they point to the frog’s current position rather than starting position in determining the frog’s next move.

Enter the formula **=I24*IF(H25="forest",1,IF(RAND()>\$G\$4,1,0))** in cell I25.

This formula repeats the formula in cell I24, except that it multiplies the result by the value of the cell above it. Thus, if our frog had died in a previous step, “safe cross” would equal 0, and cell I25 would also be 0.

Your spreadsheet should look something like the one shown in Figure 7, although your results will differ due to the random assignment of habitat cells and resulting dispersal decisions. Figure 7 shows that our frog started in cell F20, then moved to forest habitat in cell F19, then moved to forest habitat in cell G18. In the next step (step 3), there are no habitat cells present to move into, so our frog moved directly ahead into G17, thus facing a mortality risk. In this example, the random number drawn was greater than the mortality risk given in cell G4, so our frog lived through step 3 and “safe cross” is set to 1 in cell I26.

	A	B	C	D	E	F	G	H	I	J	K
8	Random landscape										
9	0.986	0.521	0.02335	0.55864	0.613	0.01877	0.5351492	0.9115978	0.013353968	0.360643762	0.0937
10	0.542	0.439	0.40328	0.873121	0.129	0.72759	0.8945317	0.40761992	0.446524093	0.524441613	0.7524
11	0.593	0.124	0.45502	0.963089	0.713	0.25849	0.8820225	0.55418912	0.902710513	0.731276432	0.0772
12	0.067	0.328	0.08485	0.505857	0.23	0.25265	0.4718367	0.1197376	0.154291029	0.503517984	0.2614
13	0.87	0.765	0.91411	0.69573	0.977	0.78992	0.0143825	0.46212178	0.709342187	0.660946157	0.3635
14	0.463	0.289	0.85953	0.626276	0.865	0.68299	0.8269911	0.09103209	0.237460091	0.306882297	0.1024
15	0.508	0.289	0.58521	0.607504	0.504	0.8257	0.3668636	0.17879266	0.134027976	0.500078004	0.4774
16	0.761	0.588	0.46878	0.14028	0.905	0.12645	0.0441818	0.12042228	0.878490401	0.865551164	0.7893
17	0.027	0.48	0.00383	0.815824	0.42	0.58778	0.7920502	0.9366276	0.830954427	0.014753477	0.1052
18	0.598	0.004	0.81365	0.061985	0.965	0.69687	0.1301638	0.54988009	0.583687251	0.608202302	0.6038
19	0.309	0.322	0.67825	0.32793	0.816	0.40047	0.2106536	0.04030575	0.054124806	0.366131204	0.771
20						Start					
21											
22	DIRECTIONS			POSITION				HABITAT CROSSED			
23	Step	Row	Column	Value	Row	Column	Address	Habitat	Safe cross?	Steps to failure	
24	1	-1	0	0.400468	19	6	\$F\$19	forest	1	#N/A	
25	2	-1	1	0.130164	18	7	\$G\$18	forest	1		
26	3	-1	0	0.79205	17	7	\$G\$17	non-forest	1		
27	4	-1	0	0.044182	16	7	\$G\$16	forest	1		

Figure 7

C. Run 25 trials of random walk simulations.

1. Set up new headings as shown in Figure 8.

2. Set up a linear series from 1 to 25 in cells A39–A63.

3. Go to Tools | Options | Calculate and click on Manual.

4. Record a macro to track 25 trials the number of times your frog successfully dispersed for 12 consecutive steps, and if it failed, the number of steps until failure (mortality).

5. In cells D39–D63, enter a formula that returns a period if cell C39 has an error code.

How did your frog do? Did it make it through 12 dispersal events with success? If not, at what step did failure occur? Press F9, the calculate key, to generate new random numbers and hence a new attempt to cross the habitat. How did your frog do in this trial? In order for us to get an appreciation of how landscape pattern (proportion of the landscape in forested habitat) and mortality risk affect dispersal abilities, we need to run several trials and obtain statistics that indicate, on average, how well our frog will fare in a given landscape. We'll run 25 trials for a given landscape proportion (e.g., 0.5 forest habitat, 0.5 nonforest habitat), and keep track of our trial results in cells A39–D63. Enter 1 in cell A39.

	B	C	D
37		# of steps to failure	
38	Success?	Raw	Refined
39	1		

Figure 8

Enter =1+A39 in cell A40. Copy this formula down to cell A63.

You need to set calculation to manual before running the macro because otherwise, the spreadsheet will recalculate between copying cell I35 and copying cell J24, and the two values will not be related.

Bring your spreadsheet macro into record mode and assign a name and shortcut key. Enter the following steps:

- Press F9 to generate new random numbers. This will change your landscape as well as the random walk steps your frog will take.
- Select cell I35, then go to Edit | Copy.
- Select cell B38. Go to Edit | Find. Leave the Find What box blank and search by columns. Select Find Next, then Close.
- Go to Edit | Paste Special, then paste in the values (not the formulae!).
- Select cell J24, then go to Edit | Copy.
- Select cell C38. Go to Edit | Find. Leave the Find What box blank and search by columns. Select Find Next, then Close.
- Go to Edit | Paste Special, then paste in the values (not the formulae!).
- Go to Tools | Macro | Stop Recording. Switch back to automatic calculation.

Now when you press your shortcut key 25 times, your results will automatically be recorded in the appropriate cells. Occasionally, your frog may disperse off the landscape. In this case, the spreadsheet will return #REF! to indicate that it cannot track the dispersal. Delete trials where this occurs until 25 valid results appear. Additionally, in trials where dispersal was successful after 12 steps, the number of steps to failure will be #N/A; that's fine.

Enter the formula =IF(ISERROR(C39),".",C39) in cell D39. Copy this formula down to cell D63.

For the trials that resulted in a failure (death), we want to know the average number of steps until death occurred. However, we cannot compute this average if any of the values in a range of cells is equal to #N/A. The formula in cell D39 takes care of this. It is an IF formula with an ISERROR formula embedded within it. The function determines if cell C39 is an error code (#N/A), and if so, the spreadsheet returns a period to indicate a missing value; if cell C39 is not an error code, the spreadsheet returns the value in cell C39.

6. In cells B65 and D65, compute the average number of trials that resulted in successful dispersal, and for failed dispersals, the average number of steps to failure.

7. Save your work.

D. Record observed and predicted rates of successful dispersal for the various proportions of habitat.

1. Set up new headings as shown in Figure 9.

2. Record the average proportion of successes in cells H39–H49 for a given landscape composition.

3. In cells I39–I49, enter a formula to compute the predicted proportion of successful dispersals in 25 trials.

4. In cells J40–J49, enter a formula to compute the expected number of steps to failure.

5. Save your work.

Enter the formula `=AVERAGE(B39:B63)` in cell B65.

Enter the formula `=AVERAGE(D39:D63)` in cell D65.

The B65 formula provides data on the probability that dispersal will be successful after 12 steps, given the landscape composition and the mortality risk for crossing into non-habitat. The D65 formula provides information on the average number of steps to failure when dispersal is not successful.

	F	G	H	I	J
37	Proportion of habitat		Proportion of successes		Predicted #
38	Forest	Nonforest	Observed	Predicted	steps to failure
39	1	0			
40	0.9	0.1			
41	0.8	0.2			
42	0.7	0.3			
43	0.6	0.4			
44	0.5	0.5			
45	0.4	0.6			
46	0.3	0.7			
47	0.2	0.8			
48	0.1	0.9			
49	0	1			

Figure 9

For example, record the value obtained in cell B65 into cell H44 (which gives the average successes when the landscape consists of 50% forest habitat and the mortality risk for crossing into nonhabitat is 0.2). Then erase your macro results, change the proportions in cells C4–C5, and run your macro again. Record the proportion of successes in the appropriate location. Then repeat the process until your table (Figure 7, column H) is complete.

Enter the formula `=(1-G4*(G39^3))^11` in cell I39. Copy this formula down to cell I49. This corresponds to Equation 1, $p(n) = [1 - R(1 - H)^3]^n$. It gives the probability that the frog will survive 11 moves across a random landscape, $p(11)$. Recall that cell **\$G\$4** holds the risk associated with crossing nonforest habitat (R), and cell G39 holds the proportion of nonforest habitat, which equals the value $1 - H$. We use 11 moves rather than 12 because the last move the frog makes is always a safe move into habitat (the pond), so it doesn't influence the overall probability of crossing the landscape safely.

Enter the formula `=1/(G4*(G40^3))` in cell J40. Copy this formula down to cell J49. This corresponds to Equation 2, $E = 1/[R(1 - H)^3]$. Once again, note that cell **\$G\$40** holds the proportion of nonforest habitat, which equals $1 - H$. We leave cell J39 blank because Equation 2 is undefined when $H = 1$.

E. Create graphs.

1. Graph the observed proportion of successes (cells H39–H49) and the expected proportion of successes (cells I39–I49) as a function of landscape composition (cells F39–F49).

Your results should resemble Figure 10, though your observed values will most likely be different.




Figure 10

2. Graph the predicted number of steps to death (cells J40–J49) as a function of proportion of forest habitat in the landscape (cells F40–F49).

Your graph should resemble Figure 11.




Figure 11

3. Save your work.

QUESTIONS

1. Fully interpret your model results. What is the nature of the average number of successes as a function of the proportion of habitat? Might this result be of interest to land planners?
2. What is the nature of the average number of steps to failure as a function of the proportion of habitat? What factors determine this result? The observed number of steps to failure calculated in cell D65 probably won't match the predicted number of steps to failure because the spreadsheet doesn't take into account the trials that did not result in failure. Can you think of ways to adjust the spreadsheet to correct this? If a landscape is "hostile" enough that an organism is not likely to survive even 12 steps, we would expect the average of the observed results to approximate the prediction. Test this by setting cell G4 to 0.7 (a high mortality rate) and running your macro 25 times each for landscapes with low proportions of forest habitat, such as 0.2, 0.1, and even 0. Compare the averages of your results to the predictions in column J.
3. How does increased mortality risk affect your results? Change cell G4 to a higher mortality risk (such as 0.5). Repeat your macro under varying landscape compositions, and record your results in a new section on your spreadsheet. Compare the dispersal success for different mortality risks in crossing nonhabitat under varying landscape scenarios.
4. How does the presence of a road in the landscape influence the proportion of successes? Clear any previous results from cells B39–C63 and run your macro 25 times with the proportion of forest set at 0.8 (cell C4) and the risk of crossing nonforest habitat at 0.5 (cell G4). Record the average number of successes. Then enter the number 1 into cells A11–K11 to simulate a road across the landscape at row 11. Clear your previous results and run your macro another 25 times. What was the influence of the road on this landscape? Now suppose that you can put in three underpasses that will allow the frogs and other species to cross the road safely. Where should the underpasses be located? Enter a 0 into the cells where an underpass is located and run your model again. Change the underpass locations and compare your results.
5. The frog in your model follows a specific algorithm in making its dispersal steps. How would your results change if you made changes to your dispersal algorithm? Try changing cells B24–B33 to vary between -1 and -2 with the **RANDBETWEEN** function. This will allow our frog to "leap" over cells in a dispersal step (this new algorithm may be better suited for a bird). Try other changes in your algorithm as well and explore its consequences on dispersal success and number of steps to death.
6. Read and discuss the papers listed under "Further Reading." Compare and contrast your model to their models and/or field data.

FURTHER READING

- Bolger, D., A. C. Alberts, R. M. Sauvajot, P. Potenza, C. McCalvin, D. Tran, S. Mazzoni and M. E. Soulé. 1997. Response of rodents to habitat fragmentation in coastal southern California. *Ecological Applications* 7: 552–563.
- With, K. and A. King. 1999. Dispersal success on fractal landscapes: A consequence of lacunarity thresholds. *Conservation Biology* 13: 314–326.
- With, K., S. J. Cadaret and C. Davis. 1999. Movement responses to patch structure in experimental fractal landscapes. *Ecology* 80: 1340–1353.